
 
*********************************************************************************
******  Do file for Naurin, Naurin and Alexander, "Gender stereotyping and chivalry in international negotiations. A survey experiment in the Council of the European Union Gender stereotypes, International Organization
*********************************************************************************
 
 
 use "Genderstereotypes.dta"

 
 **********************************************************
 *********     Main  Analyses      ************************
 **********************************************************

 *install coefplot package
 ssc install coefplot, replace

************  ttests  *****************

oneway cooperate treatment, bonferroni tabulate

ttest cooperate if treatment ==1, by(gender)
ttest cooperate if treatment ==2, by(gender)
ttest cooperate if treatment ==3, by(gender)

ttest cooperate if treatment !=2 & gender ==0, by(treatment)
ttest cooperate if treatment !=2 & gender ==1, by(treatment)



************  Figure 1  *****************

regress cooperate ibn.treatment if gender==0, nocons
estimates store male
regress cooperate ibn.treatment if gender==1, nocons
estimates store female
coefplot  (male, label(Men) msymbol(T) ciopts(lcolor(black)) mcolor(black)) ///
(female, label(Women) ciopts(lcolor(black)) mcolor(black))  ///
,xlabel(0(2)10) level(90) ///
graphregion(color(white)) 



************  Table 1  *****************

*Model 1
reg cooperate i.gender if treatment==1, robust

*Model 2   
reg cooperate i.gender c.mssizepop c.qmv if treatment==1, robust

*Model 3
reg cooperate i.gender c.gender_indexHumbert2012 c.gender_indexHumbert2012#i.gender if treatment==1, robust
margins, dydx (gender) at (gender_indexHumbert2012=(35 40 45 50 55 60 65 70 75))
marginsplot, level(90) recastci(rarea) scheme(s1mono) yline(0) ytitle("Marginal effect of gender") xtitle("Gender Equality Index") title("")

*Model 4
reg cooperate i.gender c.gender_indexHumbert2012 c.gender_indexHumbert2012#i.gender c.mssizepop c.qmv if treatment==1, robust
margins, dydx (gender) at (gender_indexHumbert2012=(35 40 45 50 55 60 65 70 75))
marginsplot, level(90) recastci(rarea) scheme(s1mono) yline(0) ytitle("Marginal effect of gender") xtitle("Gender Equality Index") title("")




************  Figure 2  *****************
graph dot gender_indexHumbert2012, linetype(line) over(ms, sort(gender_indexHumbert2012) descending) yscale(range(20 80)) ylab(20(20)80) exclude0  yline(54) scheme(s1mono) ytitle("Gender Equality Index")
graph copy Graph Figure3b, replace

reg cooperate i.gender c.gender_indexHumbert2012 c.gender_indexHumbert2012#i.gender c.mssizepop c.qmv if treatment==1, robust
margins, dydx (gender) at (gender_indexHumbert2012=(35 40 45 50 55 60 65 70 75))
marginsplot, level(90) recastci(rarea) scheme(s1mono) yline(0) ytitle("Marginal effect of gender") xtitle("Gender Equality Index") title("")
graph copy Graph Figure3a, replace

graph combine Figure3a Figure3b, scheme(s1mono)




 ***********************************************************************
 *********       Additional tests (Appendix)     ***********************
 ***********************************************************************

 

*******  In model 4 of Table 1: Substituting member state size for Big3 (Germany, France, UK = 1, All other ms = 0)  ******

gen big3 = ms
recode big3 (1/2=0) (4/5=0) (7/14=0) (16/28=0) (3 6 15 = 1) 

graph dot gender_indexHumbert2012, linetype(line) over(ms, sort(gender_indexHumbert2012) descending) yscale(range(20 80)) ylab(20(20)80) exclude0  yline(53) scheme(s1mono) ytitle("Gender Equality Index")
graph copy Graph FigureA1b, replace

reg cooperate i.gender c.gender_indexHumbert2012 c.gender_indexHumbert2012#i.gender c.big3 c.qmv if treatment==1, robust
margins, dydx (gender) at (gender_indexHumbert2012=(35 40 45 50 55 60 65 70 75))
marginsplot, level(90) recastci(rarea) scheme(s1mono) yline(0) ytitle("Marginal effect of gender") xtitle("Gender Equality Index") title("")
graph copy Graph FigureA1a, replace

graph combine FigureA1a FigureA1b, scheme(s1mono)



*******  In model 4 of Table 1: Substituting the Gender Equality Index of Humbert et al (2016) for the European Union Gender Equality Index of Plantenga et al (2009)  *******
 
gen gender_indexPlantenga = ms 
recode gender_indexPlantenga (1=61) (2=69) (3=59) (4=26) (5=37) (6=56) (7=44) (8=41) ///
(9=53) (10=65) (11=52) (12=55) (13=74) (14=72) (15=56) (16=55) (17=60) (18=59) ///
 (19=53) (20=51) (21=47) (22=56) (23=55) (24=32) (25=33) (26=.) (27=.) (28=.)
 
 preserve
 drop if gender_indexPlantenga == .
 
 
graph dot gender_indexPlantenga, linetype(line) over(ms, sort(gender_indexPlantenga) descending) yscale(range(20 80)) ylab(20(20)80) exclude0  yline(64) scheme(s1mono) ytitle("EU Gender Equality Index")
graph copy Graph FigureA2b, replace

reg cooperate i.gender c.gender_indexPlantenga c.gender_indexPlantenga#i.gender c.mssizepop c.qmv if treatment==1, robust
margins, dydx (gender) at (gender_indexPlantenga=(35 40 45 50 55 60 65 70 75))
marginsplot, level(90) recastci(rarea) scheme(s1mono) yline(0) ytitle("Marginal effect of gender") xtitle("EU Gender Equality Index") title("")
graph copy Graph FigureA2a, replace

graph combine FigureA2a FigureA2b, scheme(s1mono)

restore


*******  In model 4 of Table 1: Substituting the Gender Equality Index of Humbert et al (2016) for World Values Survey data on Gender equality (Welzel 2013)  *******
 
gen gender_indexWVS = ms 
recode gender_indexWVS (1=.) (2=.) (3=75) (4=.) (5=75) (6=79) (7=.) (8=68) ///
(9=.) (10=78) (11=.) (12=.) (13=76) (14=86) (15=72) (16=65) (17=54) (18=55) ///
 (19=62) (20=54) (21=47) (22=56) (23=74) (24=67) (25=.) (26=63) (27=58) (28=58)
 
 preserve
 drop if gender_indexWVS == .
 
graph dot gender_indexWVS, linetype(line) over(ms, sort(gender_indexWVS) descending) yscale(range(20 90)) ylab(20(20)90) exclude0  yline(69) scheme(s1mono) ytitle("WVS Gender Egalitarian Values")
graph copy Graph FigureA3b, replace

reg cooperate i.gender c.gender_indexWVS c.gender_indexWVS#i.gender c.mssizepop c.qmv if treatment==1, robust
margins, dydx (gender) at (gender_indexWVS=(45 50 55 60 65 70 75 80 85 90))
marginsplot, level(90) recastci(rarea) scheme(s1mono) yline(0) ytitle("Marginal effect of gender") xtitle("WVS Gender Egalitarian Values") title("")
graph copy Graph FigureA3a, replace

graph combine FigureA3a FigureA3b, scheme(s1mono)

restore


 ************  Multivariate test of H1 and H2  *****************

gen treatment13 = treatment
recode treatment13 (1=0) (2=.) (3=1)

reg cooperate treatment13 mssizepop qmv if gender==0, robust
reg cooperate treatment13 mssizepop qmv if gender==1, robust



 ************  Randomization check  *****************

oneway gender treatment if cooperate !=., tabulate
oneway gender_indexHumbert2012 treatment if cooperate !=., tabulate
oneway mssizepop treatment if cooperate !=., tabulate
oneway qmv treatment if cooperate !=., tabulate
 


